NASA TECHNICAL REPORT

NASA TR R-379

c. 1

NASA TR R-379

A NUMERICAL METHOD OF CALCULATING THE BOUNDARY-INDUCED INTERFERENCE IN SLOTTED OR PERFORATED WIND TUNNELS OF RECTANGULAR CROSS SECTION

by James D. Keller and Ray H. Wright

Langley Research Center
Hampton, Va. 23365

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • NOVEMBER 1971

1. Report No. NASA TR R-379	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle A NUMERICAL METHOD OF CALCULATING THE BOUNDARY-INDUCED INTERFERENCE IN SLOTTED OR PERFORATED WIND TUNNELS OF RECTANGULAR CROSS SECTION		5. Report Date November 1971
		6. Performing Organization Code
7. Author(s) James D. Keller and Ray H. Wright		8. Performing Organization Report No. L-7913
9. Performing Organization Name and Address NASA Langley Research Center Hampton, Va. 23365		10. Work Unit No. 136-13-01-05
		11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, D.C. 20546		13. Type of Report and Period Covered Technical Report
		14. Sponsoring Agency Code
15. Supplementary Notes		

16. Abstract

A numerical method is presented for calculating the boundary-induced interference at subsonic speeds in wind tunnels of rectangular cross section with slotted or perforated walls. The slot width or wall porosity can vary throughout the test section. The interference can be computed at any point in the test section. The model can be any configuration and can be located anywhere in the test section. Several examples are given, and comparison is made with other methods where available.

17. Key Words (Suggested by Author(s)) Wind-tunnel interference, Slotted wind tunnels, Perforated wind tunnels	18. Distribution Statement Unclassified – Unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	20	$3.00

*For sale by the National Technical Information Service, Springfield, Virginia 22151

A NUMERICAL METHOD OF CALCULATING THE BOUNDARY-INDUCED INTERFERENCE IN SLOTTED OR PERFORATED WIND TUNNELS OF RECTANGULAR CROSS SECTION

By James D. Keller and Ray H. Wright
Langley Research Center

SUMMARY

A numerical method is presented for calculating the boundary-induced interference at subsonic speeds in wind tunnels of rectangular cross section with slotted or perforated walls. The slot width or wall porosity can vary throughout the test section. The interference can be computed at any point in the test section. The model can be any configuration and can be located anywhere in the test section. Several examples are given, and comparison is made with other methods where available.

INTRODUCTION

One of the major problems encountered in the design of a subsonic wind tunnel is that of determining a suitable test-section configuration to reduce the interference due to the tunnel walls. The use of ventilated (slotted or perforated) wind-tunnel walls has proven to be an effective means of accomplishing this goal (ref. 1). Theoretical methods are presently available for predicting the interference due to the tunnel walls in certain cases. These methods are limited to infinite-length test sections with constant slot width or wall porosity. Some are limited as to model size, position, and load distribution.

In this report a numerical method is described for theoretically determining the boundary-induced interference in ventilated wind tunnels of rectangular cross section. The method consists of dividing the tunnel walls into a number of rectangular elements, each of which is represented by a source distribution. The boundary conditions are satisfied at the centroid of each element. The method considers test sections of finite length and can be used to satisfy a variety of boundary conditions. The boundary conditions need not be constant; thus, varying slot width or wall porosity can be treated with this method.

Several examples are presented, and a sample computer program used in making the calculations is given in an appendix.

SYMBOLS

a	effect of one element on another
b	effect of model on element
d	distance between slot centers
L	lift
$\Delta L/L$	weighting factor for a lift element
l	slot parameter
n	direction normal to wall
R	porosity restriction factor
s	wing span
t	slot width
w_t	upwash velocity caused by tunnel walls
x,y,z	Cartesian coordinates
Γ_m	circulation of model
δ	lift interference factor
ξ,η,ζ	Cartesian coordinates
σ	source distribution strength
φ	perturbation velocity potential function
φ^*	velocity potential function for an element divided by σ for the element

Subscripts:

i	ith element
j	jth element
m	model
t	tunnel walls

ANALYSIS

General Statement of the Problem

The governing equation used in the analysis of the low-speed wind-tunnel interference is

$$\frac{\partial^2 \varphi}{\partial x^2} + \frac{\partial^2 \varphi}{\partial y^2} + \frac{\partial^2 \varphi}{\partial z^2} = 0 \tag{1}$$

where φ is the perturbation velocity potential function for the entire flow field. Let $\varphi = \varphi_m + \varphi_t$ where φ_m is the potential function of the disturbances due to the model in free air and φ_t is the potential function of the additional flow due to the tunnel walls. If φ_m is taken to be a known solution of equation (1) which approximates the flow field at a distance from the model in free air, then φ_t can be determined by the fact that φ must satisfy certain boundary conditions at the tunnel walls. The objective in determining φ_t is to be able to calculate the change in the free-stream conditions caused by the tunnel walls. Since φ_m needs to be known only on the tunnel walls, any inaccuracies in ϕ_m near the model will have a negligible effect on the determination of ϕ_t.

Boundary Conditions

The boundary condition to be satisfied at a solid wall is that there can be no flow through the wall, that is

$$\frac{\partial \varphi}{\partial n} = 0$$

where n is the direction normal to the wall. The boundary condition to be satisfied at an open jet boundary is that there is no pressure difference across the boundary. This boundary condition can be approximated by (ref. 2)

$$\frac{\partial \varphi}{\partial x} = 0$$

Reference 3 shows that the mixed open and closed boundary conditions for a tunnel wall with several longitudinal slots can be replaced by a homogeneous boundary condition for an ideal slotted wall. This boundary condition is

$$\varphi + l\frac{\partial\varphi}{\partial n} = 0 \tag{2}$$

where l is a slot parameter given by

$$l = \frac{d}{\pi} \ln \csc\left(\frac{\pi}{2}\frac{t}{d}\right)$$

where t is the slot width and d is the distance between slot centers. Reference 4 gives the boundary condition for an ideal perforated wall as

$$\frac{\partial\varphi}{\partial x} + \frac{1}{R}\frac{\partial\varphi}{\partial n} = 0 \tag{3}$$

where R is an experimentally determined restriction factor.

Representation of the Tunnel Walls

In order to satisfy the homogeneous boundary conditions for ventilated (slotted or perforated) wind tunnels, the tunnel walls are divided into longitudinal strips and each strip is divided into a number of rectangular elements. The boundary conditions are satisfied at the centroid of each element. The coordinate system used has the X-axis extending along the tunnel center line, with the positive direction being the tunnel stream direction. The Z-axis is positive upward, and the Y-axis is chosen so that the coordinate system is a right-handed system. Each tunnel wall element is represented by a constant-strength source distribution over the element. If φ^* is the velocity potential function for a particular element divided by the source strength σ for that element, then

$$\varphi_t = \sum_{j=1}^{N} \varphi_j^* \sigma_j$$

where N is the total number of elements.

Consider first an element in the top or bottom wall with corners located as shown in figure 1. The potential function at a point (x,y,z) due to this source distribution is

$$\varphi^* = -\int_{\xi_1}^{\xi_2}\int_{\eta_1}^{\eta_2} \frac{d\eta\, d\xi}{\sqrt{(x-\xi)^2 + (y-\eta)^2 + \left(z-\zeta_1\right)^2}} \tag{4}$$



Figure 1.- Schematic of an element in top or bottom wall.

In order to satisfy the boundary conditions, this potential function and its derivatives must be evaluated. For convenience in writing the equations, let $x_1 = (x - \xi_1)$, $x_2 = (x - \xi_2)$, $y_1 = (y - \eta_1)$, $y_2 = (y - \eta_2)$, and $z_1 = (z - \zeta_1)$. The required equations are then

$$\begin{aligned}\varphi^* = {} & x_2 \ln\left(\frac{y_1 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{y_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right) + x_1 \ln\left(\frac{y_2 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right) \\ & + y_2 \ln\left(\frac{x_1 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right) + y_1 \ln\left(\frac{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right) \\ & + |z_1| \left\{\tan^{-1}\left[\frac{x_2 y_2}{\sqrt{z_1^2\left(x_2^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_1 y_2}{\sqrt{z_1^2\left(x_1^2 + y_2^2 + z_1^2\right)}}\right]\right. \\ & \left. - \tan^{-1}\left[\frac{x_2 y_1}{\sqrt{z_1^2\left(x_2^2 + y_1^2 + z_1^2\right)}}\right] + \tan^{-1}\left[\frac{x_1 y_1}{\sqrt{z_1^2\left(x_1^2 + y_1^2 + z_1^2\right)}}\right]\right\}\end{aligned} \tag{5}$$

$$\frac{\partial\varphi^*}{\partial x} = \ln\left[\left(\frac{y_1 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{y_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right)\left(\frac{y_2 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{y_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right)\right] \tag{6}$$

$$\frac{\partial\varphi^*}{\partial y} = \ln\left[\left(\frac{x_1 + \sqrt{x_1^2 + y_2^2 + z_1^2}}{x_2 + \sqrt{x_2^2 + y_2^2 + z_1^2}}\right)\left(\frac{x_2 + \sqrt{x_2^2 + y_1^2 + z_1^2}}{x_1 + \sqrt{x_1^2 + y_1^2 + z_1^2}}\right)\right] \tag{7}$$

$$\begin{aligned}\frac{\partial\varphi^*}{\partial z} = \operatorname{sgn}(z_1)\Bigg\{&\tan^{-1}\left[\frac{x_2y_2}{\sqrt{z_1^2\left(x_2^2 + y_2^2 + z_1^2\right)}}\right] - \tan^{-1}\left[\frac{x_1y_2}{\sqrt{z_1^2\left(x_1^2 + y_2^2 + z_1^2\right)}}\right] \\ &- \tan^{-1}\left[\frac{x_2y_1}{\sqrt{z_1^2\left(x_2^2 + y_1^2 + z_1^2\right)}}\right] + \tan^{-1}\left[\frac{x_1y_1}{\sqrt{z_1^2\left(x_1^2 + y_1^2 + z_1^2\right)}}\right]\Bigg\}\end{aligned} \tag{8}$$

In order to find the effect of an element in a side wall, it is necessary to interchange y and z in equations (5) to (8).

Computation of Source Strength Slopes

In order to compute the source strength slopes required to satisfy the boundary conditions at the centroid of each element, a matrix equation is needed which expresses these boundary conditions. Let a_{ij} be the effect at the centroid of the ith element due to the source distribution corresponding to the jth element $\left(a = \varphi^* + l\,\frac{\partial\varphi^*}{\partial n}\ \text{or}\ a = \frac{\partial\varphi^*}{\partial x} + \frac{1}{R}\,\frac{\partial\varphi^*}{\partial n}\right)$. Let b_i be the effect of the model at the centroid of the ith element $\left(b = \varphi_m + l\,\frac{\partial\varphi_m}{\partial n}\ \text{or}\ b = \frac{\partial\varphi_m}{\partial x} + \frac{1}{R}\,\frac{\partial\varphi_m}{\partial n}\right)$. Then the matrix equation that expresses the boundary conditions is

$$A\Sigma = -B \tag{9}$$

where

$$A = \left[a_{ij}\right]$$

$$\Sigma = \left[\sigma_j\right]$$

and

$$B = \left[b_i\right]$$

Equation (9) can be solved for the σ_j values which can then be used to compute the interference due to the tunnel walls.

EXAMPLES

As a relatively simple example, consider the lift interference due to a small lifting wing mounted in the center of a square test section. The wing is represented by a horseshoe vortex of circulation Γ_m. The span s of the horseshoe vortex is assumed to be so small that it becomes a vortex doublet located at (0,0,0). The perturbation velocity potential function φ_m due to this representation of the model is given by

$$\varphi_m = \frac{\Gamma_m s}{4\pi} \frac{z}{y^2 + z^2}\left(1 + \frac{x}{\sqrt{x^2 + y^2 + z^2}}\right)$$

so that

$$\frac{1}{\Gamma_m s} \frac{\partial \varphi_m}{\partial x} = \frac{1}{4\pi} \frac{z}{\left(x^2 + y^2 + z^2\right)^{3/2}}$$

$$\frac{1}{\Gamma_m s} \frac{\partial \varphi_m}{\partial y} = -\frac{1}{4\pi\left(y^2 + z^2\right)^2}\left[2yz + \frac{2x^3yz + 3xy^3z + 3xyz^3}{\left(x^2 + y^2 + z^2\right)^{3/2}}\right]$$

and

$$\frac{1}{\Gamma_m s} \frac{\partial \varphi_m}{\partial z} = \frac{1}{4\pi\left(y^2 + z^2\right)^2}\left[y^2 - z^2 + \frac{x^3y^2 + xy^4 - x^3z^2 - xy^2z^2 - 2xz^4}{\left(x^2 + y^2 + z^2\right)^{3/2}}\right]$$

These quantities are used on the right-hand side of equation (9) which is then solved for $\sigma_j/\Gamma_m s$. The $\sigma_j/\Gamma_m s$ values are suitable for the computation of the upwash velocity $\left(\frac{w_t}{\Gamma_m s} = \frac{1}{\Gamma_m s} \frac{\partial \varphi_t}{\partial z}\right)$ at any point in the test section by summing the velocities due to each element. The lift interference factor (ref. 3) is then

$$\delta = \frac{1}{2} \frac{w_t}{\Gamma_m s}$$

Each tunnel wall is divided into four strips of equal width and each strip is divided into 10 elements by cutting planes at $x = -1.00, -0.70, -0.45, -0.25, -0.10, 0.00, 0.10, 0.25, 0.45, 0.70, 1.00$. (For convenience, the tunnel width is taken to be unity.) In order to compare the results with those obtained by using the method of reference 5, let the test section have four equally spaced slots in the top and bottom walls and let the side walls be solid. Figure 2 shows the lift interference factor at the center of the tunnel as a function



Figure 2.- Lift interference factor for a small-span wing in a square tunnel with four slots in the top and bottom walls.

of the ratio of slot width to the distance between slot centers. Also shown are the results obtained by using the method of reference 5 for the same case. It can be seen that the agreement is excellent.

Figure 3 shows the longitudinal and lateral distributions of the lift interference factor due to the small-span model representation in a test section with $\frac{t}{d} = 0.04$.



Figure 3.- Longitudinal and lateral distributions of lift interference factor for a small-span wing in a square tunnel with slotted top and bottom walls. $\frac{t}{d} = 0.04$.

One of the important features of the present method is that it can be used when the slot width or tunnel porosity varies. Figure 4 shows the effect on the longitudinal distribution of the lift interference factor when the slot width is varied linearly in the tunnel stream direction. It can be seen from figure 4 that by properly contouring the slot width



Figure 4.- Effect of varying slot width on longitudinal distribution of lift interference factor in a square tunnel with four slots in the top and bottom walls.

it is possible to reduce the streamwise variations in the lift interference factor. This can be important in testing large models if it is desired to have nearly the same interference at the tail as at the wing.

Figures 5 and 6 show the lift interference factor for a small-span wing mounted in the center of a square tunnel with slotted side walls and solid top and bottom walls. The results obtained by using the method of reference 6 are also shown in these figures.



Figure 5.- Lift interference factor for a small-span wing in a square tunnel with four slots in each side wall.



Figure 6.- Longitudinal and lateral distributions of lift interference factor for a small-span wing in a square tunnel with slotted side walls. $\frac{t}{d} = 0.04$.

For the case of the perforated-wall wind tunnel, consider again the same arrangement of tunnel elements and the same model representation, but let all four walls be perforated so that the porosity restriction factor R is constant. Figure 7 shows the lift interference factor at the center of the tunnel as a function of the restriction factor. Also shown are the results from reference 7.



Figure 7.- Lift interference factor for a small-span wing in a square tunnel with perforated walls.



Figure 8.- Representation of sweptback wing.

The present method is also applicable to other representations for the model. For example, consider the case of a large-span sweptback wing which was presented in references 5 and 6. The wing spans 70 percent of the tunnel width and is swept back 35°. The wing is represented by lift elements located at points $P_1, P_2, \ldots, P_{10}$ on lines of 35° sweep as shown in figure 8. The coordinates and the assumed lift distribution for each of these points are given in the following table:

Point	x	y	$\Delta L/L$
P_1	0.0246	0.0351	0.1342
P_2	.0738	.1054	.1334
P_3	.1229	.1756	.1118
P_4	.1721	.2458	.0769
P_5	.2212	.3160	.0437
P_6	.0246	-.0351	.1342
P_7	.0738	-.1054	.1334
P_8	.1229	-.1756	.1118
P_9	.1721	-.2458	.0769
P_{10}	.2212	-.3160	.0437

Figure 9 shows the distribution of the lift interference factor along the span of this wing for a tunnel with slotted top and bottom walls and also for a tunnel with slotted side walls. The same comparison was shown in reference 6 and those results are also shown in figure 9.



Figure 9.- Spanwise distribution of lift interference factor for a large-span swept wing in slotted tunnels. $\frac{t}{d} = 0.06$.

DISCUSSION

The present numerical method of calculating wind-tunnel interference has the advantage of extreme simplicity. It is simple in concept since it involves only the satisfaction of the boundary conditions at discrete points by means of source and sink distributions over the elements of the boundary. The mathematics is also simple, involving nothing more than algebra, ordinary calculus, and the use of machine computer programs. On the other hand, the method has extremely broad applicability, since any model representation is acceptable and the model can be placed anywhere in the test section. A further advantage is that the boundary conditions can vary almost without limit over the test-section boundaries. This feature should be useful in designing test sections having small interference over the whole space occupied by the model, including the wing tips and the tail. Another advantage of the present method is the relatively short computer time required for a solution. The program presented in the appendix takes only about 1 minute on a Control Data 6600 computer system. This is in contrast to a run time of 15 minutes to 2 hours for the programs used in references 5 and 6. It should be pointed out that the present method requires only one matrix inversion even though the model is represented by several lift elements and the interference is computed at several locations in the test section. The methods of references 5 and 6 require that the infinite integrals be carried out for each lift element and for each point at which the interference is required. The present method could be made even faster in some cases by making use of the symmetry of the matrix to be inverted. The present development is oriented toward rectangular test sections, but it can easily be adapted to test sections of other cross-sectional shapes.

The present method has the disadvantage of not permitting the usual assumption of infinite length of the test section, but this limitation is regarded as of little importance since practical test sections are also limited in length by such considerations as power, cost, boundary-layer development, and operating convenience. In fact, the present method has the advantage of permitting the investigation of the effects of test-section length and of the effects of the upstream contraction region and the divergent diffuser entrance region. The accuracy is limited by the size of elements into which the test-section boundary is divided, but a judicious selection of size and distribution of elements (smaller elements in regions nearer the model) should give satisfactory results with a matrix no larger than can be inverted on a large computer. If the matrix is larger than can be inverted because of a longer test section or smaller elements, the corresponding simultaneous equations can still be solved by iteration methods, although it is doubtful whether the additional labor is justified because of the usual uncertainty regarding the actual boundary conditions.

RÉSUMÉ

A numerical method of calculating the boundary-induced interference in slotted or perforated wind tunnels of rectangular cross section has been presented. The method has broad applicability because it allows for a variation in the boundary conditions on the tunnel walls and because the model representation is arbitrary. The method also has the advantages of extreme simplicity and short computing time required.

Langley Research Center,
National Aeronautics and Space Administration,
Hampton, Va., October 18, 1971.

APPENDIX

SAMPLE FORTRAN PROGRAM

THIS APPENDIX CONTAINS A SAMPLE FORTRAN PROGRAM FOR COMPUTING THE LIFT INTERFERENCE FACTOR IN A WIND TUNNEL OF RECTANGULAR CROSS SECTION WITH SLOTTED OR PERFORATED WALLS. IT IS INTENDED ONLY AS A SAMPLE. MODIFICATIONS MUST BE MADE TO THE PROGRAM IN ORDER TO COMPUTE DIFFERENT CASES. THE PROGRAM WAS WRITTEN FOR USE ON CDC 6000 SERIES COMPUTERS.

```
      PROGRAM A3307(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
      DIMENSION XI(160),ETA(160),ZETA(160),XI1(160),XI2(160),ETA1(160),
     * ETA2(160),ZETA1(160),ZETA2(160),A(160,160),B(160),SIGMA(160)
      DIMENSION C1(160),C2(160),C3(160)
      DIMENSION XA(10),YA(10),WT(10)
      SGN(X)=SIGN(1.0,X)
      P(X1,X2,Y1,Y2,Z)=X2*ALOG(ABS((Y1+SQRT(X2*X2+Y1*Y1+Z*Z))/(Y2+SQRT(
     * X2*X2+Y2*Y2+Z*Z))))+X1*ALOG(ABS((Y2+SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+
     * SQRT(X1*X1+Y1*Y1+Z*Z))))+Y2*ALOG(ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(
     * X2+SQRT(X2*X2+Y2*Y2+Z*Z))))+Y1*ALOG(ABS((X2+SQRT(X2*X2+Y1*Y1+Z*Z)
     * )/(X1+SQRT(X1*X1+Y1*Y1+Z*Z))))
     * +ABS(Z)*(ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+Y2*Y2+Z*Z))-ATAN(X1*Y2/ABS(
     * Z)/SQRT(X1*X1+Y2*Y2+Z*Z))-ATAN(X2*Y1/ABS(Z)/SQRT(X2*X2+Y1*Y1+Z*Z)
     * )+ATAN(X1*Y1/ABS(Z)/SQRT(X1*X1+Y1*Y1+Z*Z)))
      S(X1,X2,Y1,Y2,Z)=X2*ALOG(ABS((Y1+SQRT(X2*X2+Y1*Y1+Z*Z))/(Y2+SQRT(
     * X2*X2+Y2*Y2+Z*Z))))+X1*ALOG(ABS((Y2+SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+
     * SQRT(X1*X1+Y1*Y1+Z*Z))))+Y2*ALOG(ABS((X1+SQRT(X1*X1+Y2*Y2+Z*Z))/(
     * X2+SQRT(X2*X2+Y2*Y2+Z*Z))))+Y1*ALOG(ABS((X2+SQRT(X2*X2+Y1*Y1+Z*Z)
     * )/(X1+SQRT(X1*X1+Y1*Y1+Z*Z))))
      DPDX(X1,X2,Y1,Y2,Z)=ALOG(ABS((Y1+SQRT(X2*X2+Y1*Y1+Z*Z))/(Y2+SQRT(
     * X2*X2+Y2*Y2+Z*Z))*(Y2+SQRT(X1*X1+Y2*Y2+Z*Z))/(Y1+SQRT(X1*X1+Y1*Y1
     * +Z*Z))))
      DPDY(X1,X2,Y1,Y2,Z)=DPDX(Y1,Y2,X1,X2,Z)
      DPDZ(X1,X2,Y1,Y2,Z)=SGN(Z)*(ATAN(X2*Y2/ABS(Z)/SQRT(X2*X2+Y2*Y2+Z*Z
     * ))-ATAN(X1*Y2/ABS(Z)/SQRT(X1*X1+Y2*Y2+Z*Z))-ATAN(X2*Y1/ABS(Z)/
     * SQRT(X2*X2+Y1*Y1+Z*Z))+ATAN(X1*Y1/ABS(Z)/SQRT(X1*X1+Y1*Y1+Z*Z)))
      RO(X)=ROO+DRO*(X-XI1(1))
      EL(X)=ALOG(1.0/SIN(RO(X)/2.*PI))/4./PI
      PI=3.1415926
```

THIS PART OF THE PROGRAM DEFINES THE TUNNEL GEOMETRY. HERE IT IS SET UP FOR A SQUARE TUNNEL OF UNIT WIDTH AND HEIGHT. EACH WALL IS DIVIDED INTO FOUR STRIPS AND EACH STRIP IS DIVIDED INTO TEN ELEMENTS. THE TUNNEL LENGTH IS TWICE ITS WIDTH.

```
      DO 1 I=1,160,10
      XI1(I)=-1.0
      XI1(I+1)=-.7
      XI1(I+2)=-.45
      XI1(I+3)=-.25
      XI1(I+4)=-.1
      XI1(I+5)=0.0
      XI1(I+6)=.1
      XI1(I+7)=.25
```

```
      XI1(I+8)=.45
      XI1(I+9)=.7
    1 CONTINUE
      DO 2 I=1,160
      XI2(I)=XI1(I+1)
    2 CONTINUE
      DO 3 I=10,160,10
      XI2(I)=1.0
    3 CONTINUE
      DO 4 I=1,40
      ETA1(I)=.5
      ETA2(I)=.5
      ETA1(I+40)=-.5
      ETA2(I+40)=-.5
      ZETA1(I+80)=.5
      ZETA2(I+80)=.5
      ZETA1(I+120)=-.5
      ZETA2(I+120)=-.5
    4 CONTINUE
      DO 5 I=1,10
      ZETA1(I)=.25
      ZETA1(I+10)=0.0
      ZETA1(I+20)=-.25
      ZETA1(I+30)=-.5
    5 CONTINUE
      DO 6 I=1,40
      ZETA1(I+40)=ZETA1(I)
      ETA1(I+80)=ZETA1(I)
      ETA1(I+120)=ZETA1(I)
    6 CONTINUE
      DO 7 I=1,80
      ZETA2(I)=ZETA1(I)+.25
      ETA2(I+80)=ETA1(I+80)+.25
    7 CONTINUE
      DO 8 I=1,160
      XI(I)=(XI1(I)+XI2(I))/2.
      ETA(I)=(ETA1(I)+ETA2(I))/2.
      ZETA(I)=(ZETA1(I)+ZETA2(I))/2.
    8 CONTINUE
      TSL=XI2(10)
```

THIS PART OF THE PROGRAM DEFINES THE WALL CHARACTERISTICS. IN THIS CASE THE SIDE WALLS ARE SOLID AND THE TOP AND BOTTOM WALLS EACH HAVE FOUR CONSTANT WIDTH SLOTS. THE OPEN RATIO OF THE SLOTTED WALLS IS 6 PERCENT.

```
      ROO=.06
      DRO=0.0
      DO 11 I=1,80
      C1(I)=0.0
      C2(I)=0.0
      C3(I)=1.0
   11 CONTINUE
      DO 12 I=81,160
      C1(I)=1.0
      C2(I)=0.0
      C3(I)=EL(XI(I))
   12 CONTINUE
```

```
      PRINT 991
      PRINT 992 ,(I,XI1(I),XI2(I),ETA1(I),ETA2(I),ZETA1(I),ZETA2(I),
     * XI(I),ETA(I),ZETA(I),I=1,160)
```

THIS PART OF THE PROGRAM COMPUTES THE INFLUENCE COEFFICIENTS, A(I,J)

```
      DO 190 I=1,160
      NWI=1
      IF(I.GT.40)NWI=2
      IF(I.GT.80)NWI=3
      IF(I.GT.120)NWI=4
      DO 180 J=1,160
      NWJ=1
      IF(J.GT.40)NWJ=2
      IF(J.GT.80)NWJ=3
      IF(J.GT.120)NWJ=4
      X1=XI(I)-XI1(J)
      X2=XI(I)-XI2(J)
      Y1=ETA(I)-ETA1(J)
      Y2=ETA(I)-ETA2(J)
      Z=ZETA(I)-ZETA(J)
      IF(NWJ.LT.3)Y1=ZETA(I)-ZETA1(J)
      IF(NWJ.LT.3)Y2=ZETA(I)-ZETA2(J)
      IF(NWJ.LT.3)Z=ETA(I)-ETA(J)
      U=DPDX(X1,X2,Y1,Y2,Z)
      IF(NWI.NE.NWJ)GO TO 110
      PHI=S(X1,X2,Y1,Y2,Z)
      V=0.0
      IF(I.EQ.J)V=-2.0*PI
      IF(NWJ.EQ.2)V=-V
      IF(NWJ.EQ.3)W=V
      IF(NWJ.EQ.4)W=-V
      GO TO 130
  110 CONTINUE
      PHI=P(X1,X2,Y1,Y2,Z)
      V=DPDY(X1,X2,Y1,Y2,Z)
      W=DPDZ(X1,X2,Y1,Y2,Z)
      IF(NWJ.GT.2)GO TO 120
      T=V
      V=W
      W=T
  120 CONTINUE
  130 CONTINUE
      IF(NWI.EQ.1)A(I,J)=C1(I)*PHI+C2(I)*U+C3(I)*V
      IF(NWI.EQ.2)A(I,J)=C1(I)*PHI+C2(I)*U-C3(I)*V
      IF(NWI.EQ.3)A(I,J)=C1(I)*PHI+C2(I)*U+C3(I)*W
      IF(NWI.EQ.4)A(I,J)=C1(I)*PHI+C2(I)*U-C3(I)*W
  180 CONTINUE
  190 CONTINUE
      PRINT 991
      PRINT 993 ,((I,J,A(I,J),A(I,J+1),A(I,J+2),A(I,J+3),A(I,J+4),
     * A(I,J+5),A(I,J+6),A(I,J+7),A(I,J+8),A(I,J+9),J=1,160,10),I=1,160)
```

THE FOLLOWING STATEMENT INVERTS THE MATRIX A AND PUTS THE INVERSE IN THE PLACE OF THE ORIGINAL MATRIX

```
      CALL MATRIX(10,160,160,0,A,160,DET)
```

APPENDIX

THIS PART OF THE PROGRAM COMPUTES THE DISTURBANCE DUE TO THE MODEL. HERE IT IS SET UP FOR A NUMBER OF VORTEX DOUBLETS LOCATED IN THE HORIZONTAL CENTER-PLANE OF THE TUNNEL. THE PROGRAM FIRST READS THE NUMBER OF LIFT ELEMENTS TO BE USED AND THEN READS THE X AND Y VALUES AND THE WEIGHTING FACTOR FOR EACH ELEMENT.

```
      READ 994 ,L
      READ 995 ,(XA(I),YA(I),WT(I),I=1,L)
      DO 200 I=1,160
      B(I)=0.0
  200 CONTINUE
      DO 299 K=1,L
      XPP=XA(K)
      YPP=YA(K)
      DO 210 I=1,40
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      V=-(2.*Y*Z+(2.*X**3*Y*Z+3.*X*Y**3*Z+3.*X*Y*Z**3)/(X*X+Y*Y+Z*Z)**1.
     * 5)/4./PI/(Y*Y+Z*Z)**2
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U+C3(I)*V)*WT(K)
  210 CONTINUE
      DO 220 I=41,80
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      V=-(2.*Y*Z+(2.*X**3*Y*Z+3.*X*Y**3*Z+3.*X*Y*Z**3)/(X*X+Y*Y+Z*Z)**1.
     * 5)/4./PI/(Y*Y+Z*Z)**2
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U-C3(I)*V)*WT(K)
  220 CONTINUE
      DO 230 I=81,120
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      W=(Y*Y-Z*Z+(X**3*Y*Y+X*Y**4-X**3*Z*Z-X*Y*Y*Z*Z-2.*X*Z**4)/(X*X+Y*Y
     * +Z*Z)**1.5)/4./PI/(Y*Y+Z*Z)**2
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U+C3(I)*W)*WT(K)
  230 CONTINUE
      DO 240 I=121,160
      X=XI(I)-XPP
      Y=ETA(I)-YPP
      Z=ZETA(I)
      PHI=Z/(Y*Y+Z*Z)*(1.0+X/SQRT(X*X+Y*Y+Z*Z))/4.0/PI
      U=Z/4./PI/(X*X+Y*Y+Z*Z)**1.5
      W=(Y*Y-Z*Z+(X**3*Y*Y+X*Y**4-X**3*Z*Z-X*Y*Y*Z*Z-2.*X*Z**4)/(X*X+Y*Y
     * +Z*Z)**1.5)/4./PI/(Y*Y+Z*Z)**2
      B(I)=B(I)+(C1(I)*PHI+C2(I)*U-C3(I)*W)*WT(K)
  240 CONTINUE
  299 CONTINUE
      PRINT 991
      PRINT 996 ,(B(I),I=1,160)
```

APPENDIX

THIS PART OF THE PROGRAM COMPUTES THE SOURCE STRENGTHS WHICH SATISFY THE BOUNDARY CONDITIONS.

```
    DO 300 I=1,160
    SIGMA(I)=0.0
300 CONTINUE
    DO 302 I=1,160
    DO 301 J=1,160
    SIGMA(I)=SIGMA(I)-A(I,J)*B(J)
301 CONTINUE
302 CONTINUE
    PRINT 991
    PRINT 996 ,(SIGMA(I),I=1,160)
```

THIS PART OF THE PROGRAM COMPUTES THE LIFT INTERFERENCE FACTOR. IT READS THE X, Y, AND Z VALUES AT WHICH THE INTERFERENCE IS TO BE COMPUTED.

```
    PRINT 997
400 CONTINUE
    DELTA1=0.0
    DELTA2=0.0
    DELTA3=0.0
    DELTA4=0.0
    READ 996 ,XC,YC,ZC
    IF(EOF,5)990,401
401 CONTINUE
    DO 410 J=1,40
    X1=XC-XI1(J)
    X2=XC-XI2(J)
    Y=YC-ETA(J)
    Z1=ZC-ZETA1(J)
    Z2=ZC-ZETA2(J)
    W=DPDY(X1,X2,Z1,Z2,Y)
    DELTA1=DELTA1+W*SIGMA(J)/2.
410 CONTINUE
    DO 420 J=41,80
    X1=XC-XI1(J)
    X2=XC-XI2(J)
    Y=YC-ETA(J)
    Z1=ZC-ZETA1(J)
    Z2=ZC-ZETA2(J)
    W=DPDY(X1,X2,Z1,Z2,Y)
    DELTA2=DELTA2+W*SIGMA(J)/2.
420 CONTINUE
    DO 430 J=81,120
    X1=XC-XI1(J)
    X2=XC-XI2(J)
    Y1=YC-ETA1(J)
    Y2=YC-ETA2(J)
    Z=ZC-ZETA(J)
    W=DPDZ(X1,X2,Y1,Y2,Z)
    DELTA3=DELTA3+W*SIGMA(J)/2.
430 CONTINUE
    DO 440 J=121,160
    X1=XC-XI1(J)
    X2=XC-XI2(J)
```

```
      Y1=YC-ETA1(J)
      Y2=YC-ETA2(J)
      Z=ZC-ZETA(J)
      W=DPDZ(X1,X2,Y1,Y2,Z)
      DELTA4=DELTA4+W*SIGMA(J)/2.
  440 CONTINUE
      DELTA=DELTA1+DELTA2+DELTA3+DELTA4
      PRINT 996 ,XC,YC,ZC,DELTA
      GO TO 400
  990 STOP
  991 FORMAT(1H1)
  992 FORMAT(I4,9F10.6)
  993 FORMAT(2I4,10F10.6)
  994 FORMAT(I2)
  995 FORMAT(3F10.6)
  996 FORMAT(10F10.6)
  997 FORMAT(39H1       X          Y          Z       DELTA)
      END
```

REFERENCES

1. Wright, Ray H.: The Effectiveness of the Transonic Wind Tunnel as a Device for Minimizing Tunnel-Boundary Interference for Model Tests at Transonic Speeds. AGARD Rep. 294, Mar. 1959.

2. Prandtl, L.: Applications of Modern Hydrodynamics to Aeronautics. NACA Rep. 116, 1921.

3. Davis, Don D., Jr.; and Moore, Dewey: Analytical Study of Blockage- and Lift-Interference Corrections for Slotted Tunnels Obtained by the Substitution of an Equivalent Homogeneous Boundary for the Discrete Slots. NACA RM L53E07b, 1953.

4. Goodman, Theodore R.: The Porous Wall Wind Tunnel. Part IV – Subsonic Interference Problems in a Circular Tunnel. Rep. No. AD-706-A-2 (Contract AF33(038)-9928), Cornell Aeronaut. Lab., Inc., Aug. 1951.

5. Wright, Ray H.; and Barger, Raymond L.: Wind-Tunnel Lift Interference on Sweptback Wings in Rectangular Test Sections With Slotted Top and Bottom Walls. NASA TR R-241, 1966.

6. Wright, Ray H.; and Keller, James D.: Wind-Tunnel Lift Interference on Sweptback Wings in Rectangular Test Sections With Slotted Side Walls. NASA TR R-344, 1970.

7. Schilling, Benferd L.; and Wright, Ray H.: Calculated Wind-Tunnel-Boundary Lift-Interference Factors for Rectangular Perforated Test Sections. NASA TN D-5635, 1970.

NASA-Langley, 1971 —— 1 L-7913

NATIONAL AERONAUTICS AND SPACE ADMISTRATION
WASHINGTON, D.C. 20546

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE $300

FIRST CLASS MAIL

POSTAGE AND FEES PAID
NATIONAL AERONAUTICS AND
SPACE ADMINISTRATION

U.S.MAIL

024 001 C1 U 01 711112 S00903DS
DEPT OF THE AIR FORCE
AF WEAPONS LAB (AFSC)
TECH LIBRARY/WLOL/
ATTN: E LOU BOWMAN, CHIEF
KIRTLAND AFB NM 87117

POSTMASTER: If Undeliverable (Section 1: Postal Manual) Do Not Ret

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION OFFICE

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546